
 

1 
FICHE 5 – SIMULATIONS 

 

Les commandes Scilab sont écrites en violet. Dans les exemples, toute ligne précédée de --> est 

une commande, les autres lignes sont les retours. 

 

Simuler 
 
Pour approcher la probabilité d’un événement, on va effectuer un grand nombre de fois l’expérience 

correspondante, et compter avec quelle fréquence l’événement est réalisé. 

On pourra remarquer ainsi que si on répète cette opération, la fréquence change (fluctuation 

d’échantillonnage). 

La plupart des expériences aléatoires peuvent être modélisées par des tirages aléatoires : 

 

• Tirages avec ordre et remise dans un ensemble de nombres 

 

Expérience Modélisation Scilab 

Lancer 3 fois un dé à 6 
faces 

Choisir 3 fois un nombre 
entier aléatoire entre 1 et 6 

-->T=tirage_entier(3,1,6) 

T  =  
    6.    5.    2.   

 

Expérience Modélisation Scilab 

A partir d’un point, aller 
vers la gauche ou vers la 
droite, 
4 fois 

Choisir aléatoirement 4 fois 
un nombre égal à -1 ou à  
+1 

-->t=2*tirage_entier(4,0,1)-1 
t = 

   1.    1.  - 1.    1.   

 

Dans une suite S de tirages, la fonction frequence permet d’obtenir la fréquence d’une valeur dans la 

suite : 

Par exemple pour avoir la fréquence d’apparition du 6 lors de 1000 lancers de dés, on fera : 

-->frequence(6,tirage_entier(1000,1,6))  

 ans  = 

    0.153   

 

• Tirages sans ordre ni remise dans un ensemble 

 

Pour définir un ensemble E, on mettra les éléments entre guillemets, et leur valeur éventuelle entre 

parenthèses. Par exemple, l’ensemble contenant 3 boules rouges numérotées 1, 2, 3 et 2 boules vertes 

numérotées 1, 2 se notera : 

E=ensemble("r(1)", "r(2)", "r(3)", "v(1)", "v(2)") 

 

Si on s’intéresse uniquement aux valeurs prises par les éléments, on utilisera la fonction valeur 

-->E=ensemble("r(1)", "r(2)", "r(3)", "v(1)", "v(2)") 

 E  = 

 {r(1),r(2),r(3),v(1),v(2)}    

-->valeur(E) 

 ans  = 

    1.    2.    3.    1.    2.  

  

Pour tirer 3 éléments dans un ensemble E, on utilise la commande : tirage_ensemble(3,E) 

-->tirage_ensemble(3,E) 

 ans  = 

 {r(1),r(3),v(1)}    

Exemples et exercices 

 
Exemple 1 : calculer la fréquence d’apparition du 6 lors de la simulation de 10 000 lancers d’un dé 

à 6 faces numérotées de 1 à 6. 

 

Algorithme 

 

Simuler 10000 lancers de dé, mettre les résultats dans S 

Mettre la fréquence du 6 dans f  

 

Scilab 

 

-->S=tirage_entier(10000,1,6); 

-->f=frequence(6,S) 

 f  = 

    0.1671   

 

Exemple 2 : Une puce se déplace sur un axe gradué : à chaque saut elle se déplace d’une unité, 

de manière aléatoire et équiprobable vers la droite ou vers la gauche. Elle part de l’origine et 

effectue une marche de 30 sauts. 

1) Proposer un algorithme donnant la position d’arrivée de la puce.  

 

Algorithme 

Mettre  0 dans x (qui sera l’abscisse de la puce) 

Pour i allant de 1 à 30 

  On choisit aléatoirement le nombre -1 ou le nombre  1 

  x est augmenté du nombre précédent 

Fin de pour 

Afficher x 

 

Scilab 

 

-->x=0; 

-->for i=1:30 

-->  x=x+2*tirage_entier(1,0,1)-1; 

-->end 

-->afficher("Position de la puce : "+string(x)) 

  

 Position de la puce : -10 

 

 Enrichir l’algorithme précédent pour donner la liste des positions d’arrivée de N marches 

aléatoires, faire dessiner les fréquences des différentes positions. 

 

Algorithme 

 

La fonction f  associe à N marches aléatoires les N positions de la puce, mises dans la liste P 

  Pour n allant de 1 à N 

    Mettre 0 dans P(n) : au départ la puce est à 0 

    Pour i allant de 1 à 30 

      On choisit aléatoirement le nombre -1 ou le nombre  1 

        P(n) est augmenté du nombre précédent 

    Fin de pour 

  Fin de pour 

Fin de fonction 

Calcul et dessin des fréquences pour N = 200 

Pour i allant de -30 à 30 

  Mettre la fréquence de la position i dans fr(i+31) Car on doit commencer à 1 

Fin de pour 

Effacer l’écran 

Tracer le diagramme en bâtons de fr 

   

Scilab 

 

-->function P=f(N) 

-->  for n=1:N 

-->    P(n)=0; 

-->    for i=1:30 

-->      P(n)=P(n)+2*tirage_entier(1,0,1)-1; 

-->    end 

-->  end 

-->endfunction 

 

Calcul et dessin des fréquences pour N = 200 

-->for i=-30:30 

-->  fr(i+31)=frequence(i,f(200)); 

-->end 

-->clf 

-->bar([-30:30],fr) 

 
 

 

Exemple 3 : J’ai dans ma poche 2 pièces de 10 centimes, 1 pièce de 20 centimes, 3 pièces de 50 

centimes, 1 pièce de 1 euro et 2 pièces de 2 euros. Quelle est la probabilité pour qu’en sortant deux 

pièces au hasard de ma poche, je puisse payer une baguette à 1 euro ? 

 

Algorithme 

 

Définir l’ensemble P des pièces de la poche  

Mettre 0 dans f 

Pour k allant de 1 à 1000 

  Tirer deux pièces au hasard, mettre dans t le couple obtenu 

  Mettre dans s la somme des valeurs des deux pièces 

  Si la somme est >= 1 alors 

    f prend la valeur f+1/1000 

  Fin de si 

Fin de pour 

Afficher f 

 

Scilab 

 

-->P=ensemble("a(0.1)","b(0.1)","c(0.2)","d(0.5)","e(0.5)","f(0.5)",.. 

"g(1)","h(2)","i(2)"); 

-->f=0; 

-->for k=1:1000 

-->  t= tirage_ensemble(2,P); 

-->  s=sum(valeur(t)); 

-->  if s >=1 then 

-->    f=f+1/1000; 

-->  end 

-->end 

-->f 

 f  = 

    0.677   

  

Exercice 1 

 

Simuler, sur 100 échantillons de taille 1000, le lancer d’une pièce de monnaie. Faire dessiner le nuage 

des 100 fréquences d’apparition de Pile. Refaire avec des échantillons de taille 10000. 

 

Exercice 2 

 

Elaborer un programme pour approcher la probabilité que dans une même classe de 30 élèves, 2 élèves 

au moins aient la même date de naissance. Définir une fonction associant au nombre d’élèves de la 

classe la fréquence de cet événement. A partir de combien d’élèves se produit-il plus d’une fois sur 

deux ? 

 


